Math 2280-2

Tuesday April 9, 2001

Playing with Fourier Series

```
> restart:
  with(plots):
```

On the interval [-Pi..Pi] the functions {1,cos(t),cos(2t),...sin(t),sin(2t),sin(3t),...} are mutually orthogonal with respect to the inner product

```
> Dot:=(f,g)->int(f(t)*g(t),t=-Pi..Pi);
```

$$Dot := (f, g) \to \int_{-\pi}^{\pi} f(t)\, g(t)\, dt$$

```
> Mag:=f->sqrt(Dot(f,f)); #computes the magnitude of
       #a vector
  Dist:=(f,g)->Mag(f-g);  #computes the ‘‘distance’’ between
       #two vectors.
  Angle:=(f,g)->arccos(Dot(f,g)/(Mag(f)*Mag(g)));
       #computes the ‘‘angle’’ between two vectors
```

$$Mag := f \to \sqrt{\mathrm{Dot}(f, f)}$$

$$Dist := (f, g) \to \mathrm{Mag}(f - g)$$

$$Angle := (f, g) \to \arccos\left(\frac{\mathrm{Dot}(f, g)}{\mathrm{Mag}(f)\, \mathrm{Mag}(g)}\right)$$

When we do the projection of a function f onto Vn, the coefficients for each basis vector are just the inner product of that basis vector with f, divided by the square of the magnitude of the vector Let’s illustrate by projecting the absolute value function onto W10:

```
> f:=t->abs(t);  #we’ll do Fourier for the absolute value function
  n:=3;   #order 3 expansion
  a:=vector(n); #cos coefficients
  b:=vector(n);  #sin coefficients
  a0:=(1/Pi)*Dot(f,1);  #order zero Fourier coefficient
>
```

$$f := \mathrm{abs}$$

$$n := 3$$

$$a := \mathrm{array}(1 .. 3, [\,])$$

$$b := \mathrm{array}(1 .. 3, [\,])$$

$$a0 := \pi$$

```
> for i from 1 to n do  #compute the projection coefficients
  b[i]:=(1/Pi)*Dot(f,t->sin(i*t));
  a[i]:=(1/Pi)*Dot(f,t->cos(i*t));
  od:

> evalm(b);#why will an even function have Fourier sine
    #coefficients all equal to zero?
```

$$[0, 0, 0]$$

The answer to the rhetorical question above is that the product of an even function with an odd function is an odd function. We already asserted that the integral of an odd function over an interval [-L,L] is always zero. By the way, could you prove that assertion?

```
> evalm(a); #We do these in class
```

$$\left[-4\frac{1}{\pi}, 0, -\frac{4}{9}\frac{1}{\pi}\right]$$

The answer to this rhetorical question is yes: the i=even cos coefficients are zero, the i=odd ones are -4 divided by the product of Pi with i^2.

```
> approx:=x->sum(b[k]*sin(k*x),k=1..n)
             +sum(a[m]*cos(m*x),m=1..n)
             +a0/2;
```

$$approx := x \to \left(\sum_{k=1}^{n} b_k \sin(k\,x)\right) + \left(\sum_{m=1}^{n} a_m \cos(m\,x)\right) + \frac{1}{2}\,a0$$

```
> approx(x); #order 3 Fourier expansion for abs(x):
```

$$-4\,\frac{\cos(x)}{\pi} - \frac{4}{9}\,\frac{\cos(3\,x)}{\pi} + \frac{1}{2}\,\pi$$

```
> almost:=plot(approx(x),x=-3*Pi..3*Pi,color=black):
  display({almost});
```



Now for the square wave, of period 2:

```
> f:=t->2*Heaviside(t)-1;
```

$$f := t \to 2\ \mathrm{Heaviside}(t) - 1$$

```
> plot(f(t),t=-1..1,color=black);
```




```
> n:=10;   #order 10 expansion
  a:=vector(n); #cos coefficients
  b:=vector(n);  #sin coefficients
  a0:=(1/1)*int(f(t),t=-1..1);  #order zero Fourier coefficient
>
> for i from 1 to n do  #compute the projection coefficients
  b[i]:=(1/1)*int(f(t)*sin(i*t*Pi),t=-1..1);
  a[i]:=(1/1)*int(f(t)*cos(i*t*Pi),t=-1..1);
  od:
```

$$n := 10$$

$$a := \text{array}(1 .. 10, [\,])$$

$$b := \text{array}(1 .. 10, [\,])$$

$$a0 := 0$$

```
> evalm(a);evalm(b);
```

$$[0, 0, 0, 0, 0, 0, 0, 0, 0, 0]$$

$$\left[4\frac{1}{\pi}, 0, \frac{4}{3}\frac{1}{\pi}, 0, \frac{4}{5}\frac{1}{\pi}, 0, \frac{4}{7}\frac{1}{\pi}, 0, \frac{4}{9}\frac{1}{\pi}, 0\right]$$

```
> approx5:=t->sum(b[k]*sin(k*t*Pi),k=1..5)
             +sum(a[m]*cos(m*t*Pi),m=1..5)
             +a0/2:
  approx9:=t->sum(b[k]*sin(k*t*Pi),k=1..9)
             +sum(a[m]*cos(m*t*Pi),m=1..9)
             +a0/2:
> plot({approx5(t),approx9(t)},t=-3..3,color=black);
```


1
0.5
0
–0.5
–1
–3
–2
–1
1
2
3
t

>